

## CHOLESKY DECOMPOSITION

### PURPOSE

Compute the Cholesky decomposition of a matrix.

### DESCRIPTION

If  $X$  is a positive definite matrix with row and column dimensions  $n$ , then  $X$  can be factored into an upper triangular matrix  $R$  (also of dimension  $n$ ) such that:

$$X = R'R$$

where  $R'$  refers to the transpose of  $R$ . Examples of positive definite matrices in statistical applications include the variance-covariance matrix, the correlation matrix, and the  $X'X$  matrix in regression. The Cholesky decomposition is a square root matrix (and the inverse square root matrix is the inverse of  $R$ ). For this reason, it is sometimes referred to as the Cholesky square root. The Cholesky decomposition is typically used in intermediate calculations rather than being of interest in itself. For example, the sample program below demonstrates the use of this decomposition in performing a canonical correlation analysis. Canonical correlation is discussed in most multivariate statistics texts. It can also be used to solve some systems of linear equations and in least squares fits (although DATAPLOT uses different techniques for these problems).

### SYNTAX

LET <mat2> = CHOLESKY DECOMPOSITION <mat1> <SUBSET/EXCEPT/FOR qualification>

where <mat1> is a matrix for which the Cholesky decomposition is to be computed;

<mat2> is a matrix where the Cholesky decomposition is saved;

and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

### EXAMPLES

LET R = CHOLESKY DECOMPOSITION A

### NOTE 1

A real matrix  $A$  is positive definite if and only if it is symmetric and the quadratic  $xAx$  is positive for all non-zero vectors  $x$ .

DATAPLOT only uses the upper half of the original matrix, so no test is made for symmetry. An error message is generated if a non-positive definite matrix is detected.

### NOTE 2

DATAPLOT uses the LINPACK routine SPOCO to compute the Cholesky decomposition. The reciprocal of the condition number is printed. This number gives an approximation of the numerical accuracy that was obtained when calculating the Cholesky decomposition. If this number is approximately  $10^{*-d}$ , then the elements of the decomposed matrix generally have  $d$  fewer significant digits than the original matrix.

### DEFAULT

None

### SYNONYMS

MATRIX CHOLESKY DECOMPOSITION

MATRIX CHOLESKY

CHOLESKY

### RELATED COMMANDS

MATRIX DETERMINANT	=	Compute a matrix determinant.
MATRIX INVERSE	=	Compute a matrix inverse.
MATRIX SOLUTION	=	Solve a system of linear equations.
CORRELATION MATRIX	=	Compute the correlation matrix of a matrix.
VARIANCE-COVARIANCE MATRIX	=	Compute the variance-covariance matrix of a matrix.
SINGULAR VALUES	=	Compute the singular values of a matrix.
SINGULAR VALUE FACT	=	Compute the singular value factorization of a matrix.
SINGULAR VALUE DECOM	=	Compute the singular value decomposition of a matrix.
TRIANGULAR SOLVE	=	Solve an upper (or lower) triangular system of linear equations.
TRIANGULAR INVERSE	=	Compute the inverse of an upper (or lower) triangular matrix.



## PROGRAM 2

```

. PERFORM A CANONICAL CORRELATION ANALYSIS. THIS EXAMPLE IS TAKEN
. FROM THE IMSL STAT/LIBRARY MANUAL. THIS MACRO CAN START FROM
. EITHER A VARIANCE-COVARIANCE MATRIX OR A CORRELATION MATRIX.
DIMENSION 200 COLUMNS
READ MATRIX S
1.0000 0.1839 0.0489 0.0186 0.0782 0.1147 0.2137 0.2742
0.1839 1.0000 0.2220 0.1861 0.3355 0.1021 0.4105 0.4043
0.0489 0.2220 1.0000 0.2707 0.2302 0.0931 0.3240 0.4047
0.0186 0.1861 0.2707 1.0000 0.2950 -0.0438 0.2930 0.2407
0.0782 0.3355 0.2302 0.2950 1.0000 0.2087 0.2995 0.2863
0.1147 0.1021 0.0931 -0.0438 0.2087 1.0000 0.0760 0.0702
0.2137 0.4105 0.3240 0.2930 0.2995 0.0760 1.0000 0.6247
0.2742 0.4043 0.4047 0.2407 0.2863 0.0702 0.6247 1.0000
END OF DATA

.
. CALCULATE MATRIX DIMENSIONS, SPLIT INTO S11, S12, S22
.
LET P1 = 6
LET P2 = 2
LET N = P1 + P2
LET P21 = P1 + 1
LET TAG = SEQUENCE 1 1 P1
LET TAG2 = SEQUENCE 1 1 P2
LET SA = MATRIX DEFINITION S1 P1 P1
LET SB = MATRIX DEFINITION S^P21 P1 P2
LET SD = MATRIX DEFINITION S^P21 N P2 P21
.
. CALCULATE  $C = (S11^{**(-1/2)})'S12(S22^{**(-1/2)})$ 
. CHOLESKY DECOMPOSITION IS THE MATRIX SQUARE ROOT
.
LET RA = CHOLESKY DECOMP SA
LET RAINV = TRIANGULAR INVERSE RA
LET RAINVT = MATRIX TRANSPOSE RAINV
LET RD = CHOLESKY DECOMP SD
LET RDINV = TRIANGULAR INVERSE RD
LET CZ = MATRIX MULTIPLY RAINVT SB
LET C = MATRIX MULTIPLY CZ RDINV
.
. USE SINGULAR VALUE DECOMPOSITION TO COMPUTE CANONICAL CORRELATIONS
. AND CANONICAL COEFFICIENTS. CALCULATE WILK'S LAMBDA.
.
LET LEV SV REV = SINGULAR VALUE DECOMPOSITION C
LET TEMPV = 1 - SV*SV FOR I = 1 1 P2
LOOP FOR K = 1 1 P2
    LET A = PRODUCT TEMPV SUBSET TAG2 >= K
    LET WILK(K) = A
END OF LOOP
SET WRITE FORMAT 8F10.5
PRINT " "; PRINT "CANONICAL CORRELATIONS AND WILKS LAMBDA"
PRINT SV WILK FOR I = 1 1 P2
.
. CALCULATE CANONICAL COEFFICIENTS
.
LET GAMMA1 = MATRIX MULTIPLY RAINV LEV
PRINT " "; PRINT "GROUP 1 CANNONICAL COEFFICIENTS"; PRINT GAMMA1
LET GAMMA2 = MATRIX MULTIPLY RDINV REV

```

```

PRINT “ “; PRINT “GROUP 2 CANNONICAL COEFFICIENTS”; PRINT GAMMA2
.
. CALCULATE CORRELATIONS BETWEEN VARIABLES AND CANONICAL VARIABLES
.
LET RA = MATRIX TRANSPOSE RA
LET RA = MATRIX MULTIPLY RA LEV
LET TEMP = MATRIX DIAGONAL SA
LET TEMP = 1/TEMP
LET SADIAG = DIAGONAL MATRIX TEMP
LET SADIAG = MATRIX MULTIPLY SADIAG RA
LET COEFR1 = MATRIX DEFINITION SADIAG1 P1 P2
PRINT “ “
PRINT “CORRELATIONS BETWEEN GROUP 1 VARIABLES AND GROUP 1 CANONICAL SCORES”
PRINT COEFR1
DELETE TEMP
LET TEMP = MATRIX DIAGONAL SD
LET TEMP = 1/TEMP
LET SDDIAG = DIAGONAL MATRIX TEMP
LET RD = MATRIX TRANSPOSE RD
LET RD = MATRIX MULTIPLY RD REV
LET COEFR2 = MATRIX MULTIPLY SDDIAG RD
PRINT “ “
PRINT “CORRELATIONS BETWEEN GROUP 2 VARIABLES AND GROUP 2 CANONICAL SCORES”
PRINT COEFR2

```

The following output is generated.

CANONICAL CORRELATIONS AND WILKS LAMBDA

```

0.60927  0.61590
0.14313  0.97951

```

GROUP 1 CANNONICAL COEFFICIENTS

```

-0.32591  0.41071  -0.79906  0.35773  -0.03171  0.05296
-0.48097  -0.34025  -0.08303  -0.76581  -0.48403  -0.13857
-0.45592  0.71825  0.62474  0.13386  -0.05554  0.03770
-0.20247  -0.68945  0.05961  0.73199  -0.33508  0.07983
-0.18388  -0.12472  -0.06403  -0.04460  1.07920  -0.22480
0.02657  -0.17370  0.05443  -0.08640  -0.02072  1.01690

```

GROUP 2 CANNONICAL COEFFICIENTS

```

-0.46418  -1.19355
-0.64202  1.10807

```

CORRELATIONS BETWEEN GROUP 1 VARIABLES AND GROUP 1 CANONICAL SCORES

```

-0.45175  0.34076
-0.73877  -0.29315
-0.67330  0.43128
-0.47686  -0.57988
-0.52987  -0.28106
-0.13187  -0.09029

```

CORRELATIONS BETWEEN GROUP 2 VARIABLES AND GROUP 2 CANONICAL SCORES

```

-0.86525  -0.50133
-0.93200  0.36246

```